Mat 1. 2-timersprøve den 9. december 2018.

JE 8.12.18

Opgave 1

```
> restart:
```

En cirkel C i den komplekse talplan er givet ved ligningen $|z-5| = 5$.
Dvs C er cirklen med centrum (5, 0) og radius 5.

Spørgsmål 1

```
> z1:=1+3*I;
```

$$z1 := 1 + 3\,\mathrm{I}$$

$|z_1 - 5| = |-4 + 3\,i| = \sqrt{16+9} = \sqrt{25} = 5.$
Heraf ses, at z_1 ligger på C.

Med Maple fås direkte

```
> abs(z1-5);
```

$$5$$

$$\frac{1}{z_1} = \frac{1}{1+3\,i} = \frac{1-3\,i}{(1+3\,i)(1-3\,i)} = \frac{1}{10} - \frac{3}{10}\,i.$$

Heraf ses, at $\mathrm{Re}(\frac{1}{z_1}) = \frac{1}{10}$.

Med Maple fås direkte

```
> evalc(1/z1);
```

$$\frac{1}{10} - \frac{3\,\mathrm{I}}{10}$$

Eller mere direkte

```
> Re(1/z1);
```

$$\frac{1}{10}$$

Spørgsmål 2

```
> z2:=5*sqrt(3)*exp(I*Pi/6);
```

$$z2 := 5\sqrt{3}\left(\frac{\sqrt{3}}{2} + \frac{\mathrm{I}}{2}\right)$$

```
> evalc(z2);
```

$$\frac{15}{2} + \frac{5\,\mathrm{I}\sqrt{3}}{2}$$

$$\frac{1}{z_2} = \frac{\overline{z_2}}{z_2\,\overline{z_2}} = \frac{1}{10} - \frac{\sqrt{3}\,i}{30}.$$

Heraf ses, at $\mathrm{Re}(\frac{1}{z_2}) = \frac{1}{10}$.

Med Maple fås direkte

```
> evalc(1/z2);
```

$$\frac{1}{10} - \frac{I\sqrt{3}}{30}$$

Eller mere direkte

```
> Re(1/z2);
```

$$\frac{1}{10}$$

Spørgsmål 3

```
> assume(x,real):assume(y,real);interface(showassumed=0):
> z:=x+I*y;
```

$$z := x + \mathrm{I}\, y$$

```
> Re(1/z);
```

$$\frac{x}{x^2 + y^2}$$

$\mathrm{Re}(\frac{1}{z}) = \frac{x}{x^2 + y^2} = \frac{1}{10} \Leftrightarrow x^2 + y^2 = 10\,x \Leftrightarrow x^2 + y^2 - 10\,x = 0 \Leftrightarrow (x-5)^2 + y^2 = 25 \Leftrightarrow$
$|x - 5 + iy| = 5 \Leftrightarrow |x + iy - 5| = 5 \Leftrightarrow |z - 5| = 5.$

Heraf ses, at ethvert tal z, som opfylder $\mathrm{Re}(\frac{1}{z}) = \frac{1}{10}$ ligger på C.

Specielt følger det så af spørgsmål 2, at z_2 ligger på C.

Opgave 2

```
> restart;with(LinearAlgebra):
```

Totalmatricen $\mathbf{T} = [\mathbf{A}|\ \mathbf{b}]$ for et inhomogent lineært ligningssystem har trappeformen

```
> trap(T):=<<1,0,0,0>|<0,1,0,0>|<-4,0,0,0>|<0,0,1,0>|<-2,1,3,
  0>|<1,0,-5,0>>;
```

$$trap(T) := \begin{bmatrix} 1 & 0 & -4 & 0 & -2 & 1 \\ 0 & 1 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & 3 & -5 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Spørgsmål 1

Det fuldstændigt reducerede lineære ligningssystem er da

$$\begin{array}{llll} x_1 \quad -4\,x_3 \quad -2\,x_5 = 1 & & x_1 = 1 + 4\,x_3 + 2\,x_5 \\ \quad x_2 \qquad\qquad + x_5 = 0 & \Leftrightarrow & x_2 = -x_5 \\ \qquad\qquad x_4 + 3\,x_5 = -5 & & x_4 = -5 - 3\,x_5 \end{array}$$

Sættes $x_3 = t_1$ og $x_5 = t_2$ så er den fuldstændige løsning på standardparameterform
$(x_1, x_2, x_3, x_4, x_5) = (1, 0, 0, -5, 0) + t_1(4, 0, 1, 0, 0) + t_2(2, -1, 0, -3, 1)$, $t_1, t_2 \in \mathbb{R}$.

Med Maple fås

```
> LinearSolve(trap(T),free=t);
```

$$\begin{bmatrix} 1+4\,t_3+2\,t_5 \\ -t_5 \\ t_3 \\ -5-3\,t_5 \\ t_5 \end{bmatrix}$$

Sættes $t_3 = t_1$ og $t_5 = t_2$ så fås igen

$(x_1, x_2, x_3, x_4, x_5) = (1, 0, 0, -5, 0) + t_1(4, 0, 1, 0, 0) + t_2(2, -1, 0, -3, 1)$, $t_1, t_2 \in \mathbb{R}$.

$f: \mathbb{R}^5 \to \mathbb{R}^4$ er lineær og givet ved, at ${}_e\mathbf{F}_e = \mathbf{A}$.

Spørgsmål 2

Af spørgsmål 1 fås.

$\mathbf{x} \in \ker(f) \Leftrightarrow f(\mathbf{x}) = \mathbf{0} \Leftrightarrow {}_e\mathbf{F}_{e\,e}\mathbf{x} = \mathbf{0} \Leftrightarrow \mathbf{A}\mathbf{x} = \mathbf{0} \Leftrightarrow \mathbf{x} = t_1(4, 0, 1, 0, 0) + t_2(2, -1, 0, -3, 1)$, $t_1, t_2 \in \mathbb{R}$.

Dvs $\ker(f) = \text{span}\{(4, 0, 1, 0, 0), (2, -1, 0, -3, 1)\}$, hvor de to frembringere er lineært uafhængige.

Heraf følger, at $\dim(\ker(f)) = 2$.

$\dim(f(\mathbb{R}^5)) = \dim(\mathbb{R}^5) - \dim(\ker(f)) = 5-2 = 3$.

Eller nemmere.

$\dim(f(\mathbb{R}^5)) = \rho({}_e\mathbf{F}_e) = \rho(\mathbf{A}) = 3$, hvilket aflæses af de 5 første søjler i trap($\mathbf{T}$).

$\dim(\ker(f)) = \dim(\mathbb{R}^5) - \dim(f(\mathbb{R}^5)) = 5-3 = 2$.

Opgave 3

```
> restart;with(LinearAlgebra):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
```

$\mathbb{R}^3$ udstyret med det sædvanlige prikprodukt.

```
> v1:=<1,1,1>;
```

$$v1 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

```
> v2:=<1,0,-1>;
```

$$v2 := \begin{bmatrix} 1 \\ 0 \\ -1 \end{bmatrix}$$

```
> v3:=<-1,1,0>;
```

$$v3 := \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$$

Spørgsmål 1

Koordinatmatricen for vektorerne **v**1, **v**2 og **v**3 med hensyn til standardbasis e er

```
> eV:=<v1|v2|v3>;
```

$$eV := \begin{bmatrix} 1 & 1 & -1 \\ 1 & 0 & 1 \\ 1 & -1 & 0 \end{bmatrix}$$

Da

```
> rho('eV')=Rank(eV);
```

$$\rho(eV) = 3$$

er **v**1, **v**2 og **v**3 tre lineært uafhængige vektorer i $\mathbb{R}^3$ og dermed er v = (**v**1, **v**2, **v**3) en basis for $\mathbb{R}^3$.

$f : \mathbb{R}^3 \rightarrow \mathbb{R}^3$ er lineær og givet ved, at der for enhver vektor $\mathbf{u} \in \mathrm{span}\{\mathbf{v}1\}$ gælder
$f(\mathbf{u}) = 5\mathbf{u}$
og for enhver vektor $\mathbf{u} \in \mathrm{span}\{\mathbf{v}2, \mathbf{v}3\}$ gælder
$f(\mathbf{u}) = -4\mathbf{u}$.

Spørgsmål 2

Af det givne aflæses, at 5 er en egenværdi for f med tilhørende egenvektorrum $E_5 = \mathrm{span}\{\mathbf{v}1\}$ og at -4 er en egenværdi for f med tilhørende egenvektorrum $E_{-4} = \mathrm{span}\{\mathbf{v}2, \mathbf{v}3\}$.
v = (**v**1, **v**2, **v**3) er således en basis for $\mathbb{R}^3$ bestående af egenvektorer for f.
Da $f(\mathbf{v}1) = 5\mathbf{v}1 = 5\mathbf{v}1 + 0\mathbf{v}2 + 0\mathbf{v}3$, $f(\mathbf{v}2) = -4\mathbf{v}2 = 0\mathbf{v}1 - 4\mathbf{v}2 + 0\mathbf{v}3$ og
$f(\mathbf{v}3) = -4\mathbf{v}3 = 0\mathbf{v}1 + 0\mathbf{v}2 - 4\mathbf{v}3$, så er

```
> vFv:=DiagonalMatrix(<5,-4,-4>);
```

$$vFv := \begin{bmatrix} 5 & 0 & 0 \\ 0 & -4 & 0 \\ 0 & 0 & -4 \end{bmatrix}$$

Basisskiftematricen, der skifter fra v-koordinater til standard e-koordinater, er

```
> eMv:=eV;
```

$$eMv := \begin{bmatrix} 1 & 1 & -1 \\ 1 & 0 & 1 \\ 1 & -1 & 0 \end{bmatrix}$$

og basisskiftematricen, der skifter fra standard e-koordinater til v-koordinater, er

```
> vMe:=eMv^(-1);
```

$$vMe := \begin{bmatrix} \frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & -\frac{2}{3} \\ -\frac{1}{3} & \frac{2}{3} & -\frac{1}{3} \end{bmatrix}$$

Vi har da

```
> eFe:=eMv.vFv.vMe;
```

$$eFe := \begin{bmatrix} -1 & 3 & 3 \\ 3 & -1 & 3 \\ 3 & 3 & -1 \end{bmatrix}$$

som ses, at være symmetrisk.

Spørgsmål 3

Da ${}_e\mathbf{F}_e$ er symmetrisk, så er de to egenvektorrum E_5 og E_{-4} ortogonale, hvilket også ses direkte.

```
> q1:=v1/sqrt(prik(v1,v1));
```

$$q1 := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix}$$

er en ortonormal basis for E_5 .

```
> q2:=v3/sqrt(prik(v3,v3));
```

$$q2 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

og

```
> q3:=kryds(q1,q2);
```

$$q3 := \begin{bmatrix} -\frac{\sqrt{3}\,\sqrt{2}}{6} \\ -\frac{\sqrt{3}\,\sqrt{2}}{6} \\ \frac{\sqrt{3}\,\sqrt{2}}{3} \end{bmatrix}$$

udgør en ortonormal basis for E_{-4} .

(**q**1, **q**2, **q**3) er da en ortonormal basis for $\mathbb{R}^3$ udstyret med det sædvanlige prikprodukt bestående af egenvektorer for f , hvor basisvektoren **q**2 er ensrettet med **v**3.

Opgave 4

```
> restart;with(LinearAlgebra):
```

Givet

$$\begin{bmatrix} x_1'(t) \\ x_2'(t) \end{bmatrix} = \begin{bmatrix} a & 3-a \\ 1 & 3 \end{bmatrix} \begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix}, \ t \in \mathbb{R},$$

hvor a er et vilkårligt reelt tal.

Systemmatrix

```
> A:=<<a,1>|<3-a,3>>;
```

$$A := \begin{bmatrix} a & 3-a \\ 1 & 3 \end{bmatrix}$$

Spørgsmål 1

```
> a:=3:
```

Systemmatricen er i dette tilfælde

```
> A;
```

$$\begin{bmatrix} 3 & 0 \\ 1 & 3 \end{bmatrix}$$

```
> g:=(x,y)->evalb(x[1]<y[1]):
> sort(Eigenvectors(A,output=list),g);
```

$$\left[\left[3, 2, \left\{\begin{bmatrix} 0 \\ 1 \end{bmatrix}\right\}\right]\right]$$

Heraf aflæses, at eneste egenværdi for systemmatricen **A** er 3 med am(3) = 2 og gm(3) = 1. Da gm(3) < am(3) kan systemmatricen **A** ikke diagonaliseres (**A** har ikke to lineært uafhængige egenvektorer) og dermed kan ligningssystemet ikke løses ved diagonaliseringsmetoden.

Spørgsmål 2

```
> a:='a':
```

```
> A;
```

$$\begin{bmatrix} a & 3-a \\ 1 & 3 \end{bmatrix}$$

Da

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = e^{(4+i)t} \begin{bmatrix} 1+i \\ 1 \end{bmatrix}$$ er en partikulær løsning til differentialligningssystemet,

haves

$$(4+i)e^{(4+i)t} \begin{bmatrix} 1+i \\ 1 \end{bmatrix} = \mathbf{A}\, e^{(4+i)t} \begin{bmatrix} 1+i \\ 1 \end{bmatrix} \Leftrightarrow (4+i) \begin{bmatrix} 1+i \\ 1 \end{bmatrix} = \mathbf{A} \begin{bmatrix} 1+i \\ 1 \end{bmatrix}.$$

Heraf aflæses, at $\lambda 1 = 4 + i$ er en egenværdi for **A** og at **v**1 = $\begin{bmatrix} 1+i \\ 1 \end{bmatrix}$ er en tilhørende egenvektor for **A**. Da **A** er en reel 2×2- matrix, så er $\lambda 2 = \overline{\lambda 1} = (4 - i)$ den anden egenværdi for **A** og **v**2 = $\overline{\mathbf{v}1}$ = $\begin{bmatrix} 1-i \\ 1 \end{bmatrix}$ er en tilhørende egenvektor for **A**.

Da **v**1 og **v**2 er to lineært uafhængige egenvektorer for **A** (de tilhørende egenværdier er forskellige), så er den fuldstændige komplekse løsning til differentialligningssystemet

$$\begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix} = c1 \mathrm{e}^{(4+i)t} \begin{bmatrix} 1+i \\ 1 \end{bmatrix} + c2\, \mathrm{e}^{(4-i)t} \begin{bmatrix} 1-i \\ 1 \end{bmatrix}, \ t \in \mathbb{R}\ , c1, c2 \in \mathbb{C}.$$

Spørgsmål 3

Da $4 + i$ er en egenværdi for systemmatricen **A**, så er $4 + i$ rod i det karakteristiske polynomium for **A**, som er

```
> P:=lambda->Determinant(A-lambda*IdentityMatrix(2)):
> simplify(P(lambda));
```

$$\lambda^2 + (-a - 3)\lambda + 4a - 3$$

```
> P(4+I);
```

$$5\,\mathrm{I} - \mathrm{I}\,a$$

Heraf ses, at $P(4 + i) = 0 \Leftrightarrow a = 5$.

Altså. Den værdi af a , som har været benyttet i spørgsmål 2, er $a = 5$.